

## Rules for integrands of the form $P_q[x] (a + b x^2)^p$

1:  $\int P_q[x] (a + b x^2)^p dx$  when  $p \in \mathbb{Z}^+ \wedge P_q[x, 1] \neq 0$

Derivation: Algebraic expansion

Rule 1.1.2.x.1: If  $p \in \mathbb{Z}^+ \wedge P_q[x, 1] \neq 0$ , then

$$\int P_q[x] (a + b x^2)^p dx \rightarrow \frac{P_q[x, 1] (a + b x^2)^{p+1}}{2 b (p + 1)} + \int (P_q[x] - P_q[x, 1] x) (a + b x^2)^p dx$$

Program code:

```
Int [Pq_* (a_+b_.*x_^2)^p_.,x_Symbol] :=
  Coeff [Pq,x,1] * (a+b*x^2)^(p+1) / (2*b*(p+1)) +
  Int [ExpandToSum [Pq-Coeff [Pq,x,1]*x,x] * (a+b*x^2)^p,x] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[p,0] && NeQ[Coeff [Pq,x,1],0]
```

2:  $\int P_q[x] (a + b x^2)^p dx$  when  $p + 2 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.2.x.2: If  $p + 2 \in \mathbb{Z}^+$ , then

$$\int P_q[x] (a + b x^2)^p dx \rightarrow \int \text{ExpandIntegrand} [P_q[x] (a + b x^2)^p, x] dx$$

Program code:

```
Int [Pq_* (a_+b_.*x_^2)^p_.,x_Symbol] :=
  Int [ExpandIntegrand [Pq*(a+b*x^2)^p,x],x] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[p,-2]
```

3:  $\int P_q[x] (a + bx^2)^p dx$  when  $\text{PolynomialRemainder}[P_q[x], x, x] == 0$

Derivation: Algebraic simplification

Rule 1.1.2.x.3: If  $\text{PolynomialRemainder}[P_q[x], x, x] == 0$ , then

$$\int P_q[x] (a + bx^2)^p dx \rightarrow \int x \text{PolynomialQuotient}[P_q[x], x, x] (a + bx^2)^p dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^2)^p_,x_Symbol] :=
  Int[x*PolynomialQuotient[Pq,x,x]*(a+b*x^2)^p,x] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,x,x],0] && Not[MatchQ[Pq,x^m_.*u_./; IntegerQ[m]]]
```

$$4: \int P_q[x^2] (a+bx^2)^p dx \text{ when } p + \frac{1}{2} \in \mathbb{Z}^- \wedge 2q + 2p + 1 < 0$$

Derivation: Algebraic expansion and binomial recurrence 3b

$$\text{Basis: } \int (a+bx^2)^p dx = \frac{x(a+bx^2)^{p+1}}{a} - \frac{b(2p+3)}{a} \int x^2 (a+bx^2)^p dx$$

Note: Interestingly this rule eliminates the constant term of  $P_q[x^2]$  rather than the highest degree term.

Rule 1.1.2.x.4: If  $p + \frac{1}{2} \in \mathbb{Z}^- \wedge 2q + 2p + 1 < 0$ , let  $A \rightarrow P_q[x^2, \theta]$  and  $Q_{q-1}[x^2] \rightarrow \text{PolynomialQuotient}[P_q[x^2] - A, x^2, x]$ , then

$$\begin{aligned} & \int P_q[x^2] (a+bx^2)^p dx \\ & \rightarrow A \int (a+bx^2)^p dx + \int x^2 Q_{q-1}[x^2] (a+bx^2)^p dx \\ & \rightarrow \frac{Ax(a+bx^2)^{p+1}}{a} + \frac{1}{a} \int x^2 (a+bx^2)^p (aQ_{q-1}[x^2] - Ab(2p+3)) dx \end{aligned}$$

Program code:

```
Int[Pq_*(a_+b_*x_^2)^p_,x_Symbol] :=
  With[{A=Coeff[Pq,x,0],Q=PolynomialQuotient[Pq-Coeff[Pq,x,0],x^2,x]},
    A*x*(a+b*x^2)^(p+1)/a + 1/a*Int[x^2*(a+b*x^2)^p*(a*Q-A*b*(2*p+3)),x] /;
  FreeQ[{a,b},x] && PolyQ[Pq,x^2] && ILtQ[p+1/2,0] && LtQ[Expon[Pq,x]+2*p+1,0]
```

$$5: \int P_q[x] (a+bx^2)^p dx \text{ when } p < -1$$

Derivation: Algebraic expansion and quadratic recurrence 2a

Rule 1.1.2.x.5: If  $p < -1$ ,

let  $Q_{q-2}[x] \rightarrow \text{PolynomialQuotient}[P_q[x], a+bx^2, x]$  and  $f + gx \rightarrow \text{PolynomialRemainder}[P_q[x], a+bx^2, x]$ , then

$$\int P_q[x] (a+bx^2)^p dx \rightarrow$$

$$\int (f + gx) (a + bx^2)^p dx + \int Q_{q-2}[x] (a + bx^2)^{p+1} dx \rightarrow$$

$$\frac{(ag - bfx) (a + bx^2)^{p+1}}{2ab(p+1)} + \frac{1}{2ab(p+1)} \int (a + bx^2)^{p+1} (2ab(p+1)Q_{q-2}[x] + bf(2p+3)) dx$$

### Program code:

```
Int [Pq_ * (a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+b*x^2,x],
    f=Coeff[PolynomialRemainder[Pq,a+b*x^2,x],x,0],
    g=Coeff[PolynomialRemainder[Pq,a+b*x^2,x],x,1]},
    (a*g-b*f*x) * (a+b*x^2)^(p+1) / (2*a*b*(p+1)) +
    1 / (2*a*b*(p+1)) * Int[(a+b*x^2)^(p+1) * ExpandToSum[2*a*b*(p+1)*Q+b*f*(2*p+3),x],x] /;
  FreeQ[{a,b},x] && PolyQ[Pq,x] && LtQ[p,-1]
```

6:  $\int P_q[x] (a + bx^2)^p dx$  when  $p \neq -1$

Reference: G&R 2.160.3

Derivation: Trinomial recurrence 3a with  $A = 0$ ,  $B = 1$  and  $m = m - n$

Reference: G&R 2.104

Note: This special case of the Ostrogradskiy-Hermite integration method reduces the degree of the polynomial in the resulting integrand.

Rule 1.1.2.x.6: If  $p \neq -1$ , let  $e \rightarrow P_q[x, q]$ , then

$$\int P_q[x] (a + bx^2)^p dx \rightarrow$$

$$\int (P_q[x] - ex^q) (a + bx^2)^p dx + e \int x^q (a + bx^2)^p dx \rightarrow$$

$$\frac{e x^{q-1} (a + b x^2)^{p+1}}{b (q + 2 p + 1)} + \frac{1}{b (q + 2 p + 1)} \int (a + b x^2)^p (b (q + 2 p + 1) P_q[x] - a e (q - 1) x^{q-2} - b e (q + 2 p + 1) x^q) dx$$

### Program code:

```
Int [Pq_* (a_+b_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],e=Coeff[Pq,x,Expon[Pq,x]]},
    e*x^(q-1)*(a+b*x^2)^(p+1)/(b*(q+2*p+1)) +
    1/(b*(q+2*p+1))*Int[(a+b*x^2)^p*ExpandToSum[b*(q+2*p+1)*Pq-a*e*(q-1)*x^(q-2)-b*e*(q+2*p+1)*x^q,x]] /;
  FreeQ[{a,b,p},x] && PolyQ[Pq,x] && Not[LeQ[p,-1]]
```